H J Fix #1

Astronuclear

MASTER

WANL-TMI-1212

August 21, 1964

Henry J. Fix

FILE COPY

HM04-A GENERAL LOSS COEFFICIENT AND FRICTION FACTOR PROGRAM FOR VARIOUS GEOMETRIES USING AIR OR HYDROGEN

by

H. L. Morrison
Fluid Flow Laboratory
Experimental Engineering

Henry J. Fix

Approved By:

Henry J. Fix, Supervisor
Fluid Flow Test & Analysis

INFORMATION CATEGORY

UNCLASSIFIED

Authorized Classifier 8/25/64 Date

DISCLAIMER

INTRODUCTION

A simple FORTRAN program has been written to compute loss coefficient or friction factor for a given geometry using experimental data. Also the Reynolds number and Mach number are computed for the given conditions. The equations defining the calculated values are given below.

The experimentally measured pressures and pressure drops may be input in the units of measurement, i.e., in psi, inches of mercury, or inches of water. All readings are converted to psi within the program.

If the geometry is one or more circular holes, a circular annulus, or a rectangular annulus, the program will compute the area and equivalent diameter given the necessary dimensions. Any other configuration must be designated by inputing its area and equivalent diameter.

The working fluid may be either air or hydrogen. The properties of hydrogen are calculated from a normal (25% para-) hydrogen properties subroutine. Air is assumed to obey the perfect gas law and to have a specific heat ratio of 1.40. The viscosity of air is given by a least squares fit to data obtained from Hunsaker and Rightmire.

The total mass flow through the system can be measured by a sonic or a reversible, adiabatic nozzle. Again the measured pressures can be input in experimental units, and are converted to psi within the program. The mass flow equations are also given later.

In all cases the temperature data must be input in degrees Fahrenheit and lengths in inches.

The program computes the required data for any number of test points taken for a given geometry. Also different problems may be input consecutively.

EQUATIONS

In the following, although the same symbol may be used in different equations, it is replaced in the program by a different variable name each time it occurs.

Loss Coefficient

$$C = \frac{2 \rho g \Delta P A^2}{144 w^2}$$

where

C = loss coefficient

ρ = density evaluated at given temperature and pressure (lb/ft^3)

g = 32.2 ft/sec^2

ΔP = pressure drop across component ($lb/in.^2$)

A = area of flow ($in.^2$)

w = mass flow rate of fluid (lb/sec)

Friction Factor

$$f = \frac{2 \rho g \Delta P A^2}{144 w^2} \frac{De}{L}$$

where

f = friction factor

De = equivalent diameter (in.)

L = length between pressure drop taps (in.)

Reynolds Number

$$Re = \frac{12 \ w \ De}{\mu A}$$

where

μ = viscosity evaluated at given temperature and pressure (lb/ft-sec)

Mach Number

$$M = \frac{144 \ w}{\rho A \sqrt{kg \ RT}}$$

where

k = specific heat ratio evaluated at given temperature and pressure

R = 53.33 ft/°R for air or 766.4 ft/°R for hydrogen

T = temperature (°R)

Mass Flow Rate

Sonic

$$w = \frac{C \ P}{\sqrt{T}}$$

where

w = mass flow rate (lb/sec)

C = orifice coefficient

P = absolute pressure upstream of orifice (lb/in.^{2})

T = temperature at nozzle (°R)

Reversible, Adiabatic

$$w = \frac{C'}{12} \left[\frac{2k \rho g P \left(R^{2/k} - R^{(k+1)/k} \right)}{(k-1)\left(1-\beta^2 R^{2/k}\right)} \right]^{\frac{1}{2}}$$

where

C' = CA

C = orifice coefficient

A = orifice area (in.^{2})

k = specific heat ratio evaluated at conditions upstream of orifice

ρ = density evaluated at conditions upstream of orifice (lb/ft^3)

P = absolute pressure upstream of orifice (lb/in.^{2})

R = ratio of absolute downstream to absolute upstream pressure

β = ratio of orifice area to pipe area

Input

The input consists of cards which describe the geometry and fluid type, the units of measurement, and other characteristics which remain the same during a particular test. The other cards give the pressure, temperature, and pressure drop for the component and the flow orifice measured during the test. There should be one of these cards for each test point.

A description of the input and a sample input sheet for each type of constant follows.

Card Number	Columns	Description	
1	1-80	Comments about the test: any	
2	1-80		characters are permitted
3	1-8	type of fluid	0. hydrogen 1. air
	9-16	type of geometry	0. circular hole (s) 1. circular annulus 2. rectangular annulus 3. other
	17-24	units of pressure	0. psi gauge 1. in. mercury 2. in. water
	25-32	units of pressure drop	0. psi gauge 1. in. mercury 2. in. water
	33-40	type of constant	0. loss coefficient 1. friction factor
	41-48	number of test points	
	49-56	barometric pressure (in. Hg)	
4	1-8	geometry information	if circular holes, give number of holes. if circular annulus, give smaller diameter. if rectangular annulus, give length. if other, give value of area.

Card Number	Columns	Description	
4	9-16	geometry information	if circular holes, give hole diameter. if circular annulus, give larger diameter. if rectangular annulus, give width. if other, give the value of equivalent diameter.
5	1-8	type of flow orifice	0. sonic 1. reversible, adiabatic
	9-16	units of orifice pressure	0. psi gauge 1. in. mercury 2. in. water
	17-24	flow constant (C for sonic, C' for reversible, adiabatic)	
	25-32	β (only needed for reversible, adiabatic; may be left blank for sonic)	
6	1-8	length between pressure drop taps (this card is needed only when calculating friction factor and <u>must</u> be omitted when loss coefficient is calculated)	
6 or 7 through number of test points	1-8	value of pressure at which properties are to be evaluated	
	9-16	value of pressure at which properties are to be evaluated	
	17-24	value of temperature at which properties are to be evaluated	
	25-32	value of temperature at which properties are to be evaluated	
	33-40	value of pressure drop across component	
	41-48	value of pressure drop across component	
	49-56	value of pressure inlet to flow measuring orifice	
	57-64	value of pressure exit from flow measuring orifice	
	65-72	value of temperature at flow measuring orifice	

In the above input all values which involve length must have dimensions of inches and all temperatures are input in degrees Fahrenheit. Further, all input values must contain a decimal point.

The two values of pressure, temperature, and pressure drop are averaged arithmetically in the program. Thus, if only one value is measured, it should be entered in both locations.

FORM 33406

Westinghouse Electric Corporation

CODING FORM FOR IBM TYPE 7090 COMPUTER
FORTRAN-FAP

TITLE	ANALYST	PHONE	DATE	PROGRAM NO.	DECK NO.	L.S. NO.
	H L. MORRISON	708	8-6-64	HM04		43634

AUG 6 3 47 PM '64

FORTRAN STAT. NO. / FAP SYMBOL (1–6)	FORTRAN STATEMENT / OPERATION (7–14)	ADDRESS, TAG, DECREMENT (15–21)	(22–29)	REMARKS (30–)				
FFL-8B	CONTROL DRUM ORIFICE			CALIBRATION OF CONTROL DRUM ANNULUS ON AIR				
RUN NO.	1 .048 INCH ANNULUS			JUNE 24, 1964 INLET LOSS COEFFICIENT				
1.	1.	0.	1.	0.	10.	29.22		
4.120	4.216							
1.	1.	0.7697	0.49421					
85.	85.5	81.	80.	2.05	2.4	1.00	0.	80.
84.5	85.	81.	81.	4.15	4.8	2.00	0.	81.
85.	85.5	83.	82.	6.05	7.2	3.05	0.	82.
85.5	85.5	81.	80.	7.75	9.15	4.05	0.	80.
84.5	85.	80.	79.	9.5	11.2	5.00	0.	79.
84.5	85.	79.	79.	11.4	12.7	6.10	0.	79.
84.5	85.	78.	78.	13.	14.35	7.05	0.	78.
85.	85.5	77.	77.	14.85	16.65	8.20	0.	77.
85.5	85.5	77.	76.	15.95	18.	9.20	0.	76.
84.5	85.	77.	77.	17.	19.25	10.00	0.	77.

Sample Input Sheet for Loss Coefficient Calculation

FORM 33406

Westinghouse Electric Corporation

CODING FORM FOR IBM TYPE 7090 COMPUTER
FORTRAN-FAP

TITLE	ANALYST	PHONE	DATE	PROGRAM NO.	DECK NO.	L.S. NO.
	H. L. MORRISON	708	8-15-64	HM04		43634

FORTRAN: STAT. NO. / FORTRAN STATEMENT
FAP: SYMBOL / OPERATION / ADDRESS, TAG, DECREMENT / REMARKS

FFL-88 CONTROL DRUM ORIFICE TEST CALIBRATION OF DRUM ANNULUS ON AIR
RUN NO. 1 .048 INCH ANNULUS JUNE 29, 1964 FRICTION FACTOR THRU ANNULUS

1.	1.	0.	1.	1.	10.	29.22		
4.120	4.275							
1.	1.	0.7697	0.49421					
51.75								
84.	84.5	81.	80.	0.7	0.25	1.00	0.	80.
82.5	82.5	81.	81.	0.9	0.8	2.00	0.	81.
82.	82.	83.	82.	1.85	1.0	3.05	0.	82.
81.5	81.	81.	80.	2.2	1.1	4.05	0.	80.
81.	79.5	80.	79.	2.7	1.45	5.00	0.	79.
79.	79.	79.	79.	3.25	2.35	6.10	0.	79.
78.5	78.	78.	78.	4.1	3.0	7.05	0.	78.
78.	77.5	77.	77.	5.3	3.8	8.20	0.	77.
77.5	76.5	77.	76.	4.8	3.4	9.20	0.	76.
76.5	75.5	77.	77.	5.3	3.5	10.00	0.	77.

Sample Input Sheet for Friction Factor Calculation

FORTRAN SOURCE PROGRAM

```
      DIMENSION COMM(40)                                                8/31/64

      DIMENSION COMM(40)
C     THIS PROGRAM CALCULATES LOSS COEFF. OR FRICTION FACTOR FOR VARIOUS
C     GEOMETRIES.  PRESSURE AND PRESSURE DROPS CAN BE IN PSI, IN. H20 OR
C     IN. HG.  ALL DIMENSIONS IN INCHES AND TEMPERATURES IN DEG. FAHR.
C     ALSO CALCULATES REYNOLDS AND MACH NUMBERS FOR GIVEN GEOMETRY
 5000 READ INPUT TAPE 5,1,(COMM(I),I=1,40)
    1 FORMAT(20A4)
      READ INPUT TAPE 5,2,FLTYPE,GEOTYP,PUNIT,DPUNIT,CONTYP,CASES,PA
    2 FORMAT(10E8.1)
      IF(GEOTYP)10000,10,11
   10 READ INPUT TAPE 5,2,HOLES,DIA
      AREA=0.7854*HOLES*DIA**2
      EQDIA=DIA
      GO TO 100
   11 IF(GEOTYP-1.1)12,10000,13
   12 READ INPUT TAPE 5,2,DIA1,DIA2
      AREA=0.7854*(DIA2**2-DIA1**2)
      EQDIA=DIA2-DIA1
      GO TO 100
   13 IF(GEOTYP-2.1)14,10000,15
   14 READ INPUT TAPE 5,2,A,B
      AREA=A*B
      EQDIA=2.0*A*B/(A+B)
      GO TO 100
   15 READ INPUT TAPE 5,2,AREA,EQDIA
  100 WRITE OUTPUT TAPE 6,3,(COMM(I),I=1,40),AREA,EQDIA
    3 FORMAT(1H1,15X,20A4//15X,20A4//15X,6HAREA =,F8.5,8X,21HEQUIVALENT
     1DIAMETER =,F8.5)
      READ INPUT TAPE 5,2,ZONTYP,PNUNIT,C,BETASQ
      IF(CONTYP)10000,17,16
   16 READ INPUT TAPE 5,2,DIS
      D=EQDIA/DIS
      WRITE OUTPUT TAPE 6,4
    4 FORMAT(/5X,4HPAVE,5X,4HTAVE,5X,5HDPAVE,3X,7HDENSITY,5X,9HVISCOSITY
     1,7X,1HK,7X,4HFLOW,7X,2HFF,7X,8HREYNOLDS,6X,4HMACH/5X,4HPSIG,5X,5HD
     2EG.F,4X,4HPSID,4X,7HLB/CUFT,5X,9HLB/FT-SEC,15X,6HLB/SEC)
      GO TO 200
   17 D=1.0
      WRITE OUTPUT TAPE 6,5
    5 FORMAT(/5X,4HPAVE,5X,4HTAVE,5X,5HDPAVE,3X,7HDENSITY,5X,9HVISCOSITY
     1,7X,1HK,7X,4HFLOW,5X,5HCLOSS,6X,8HREYNOLDS,6X,4HMACH/5X,4HPSIG,5X,
     25HDEG.F,4X,4HPSID,4X,7HLB/CUFT,5X,9HLB/FT-SEC,15X,6HLB/SEC)
  200 NCASES=CASES
```

FORTRAN SOURCE PROGRAM
(Continued)

```
      DIMENSION COMM(40)                                                8/31/64
      ATM=PA*0.4912
      DO 1000 J=1,NCASES
      READ INPUT TAPE 5,2,P1A,P1B,T1,T2,DP1,DP2,PUP,PDOWN,TN
      PBAR=(P1A+P1B)/2.0
      TBAR=(T1+T2)/2.0
      DPBAR=(DP1+DP2)/2.0
      IF(PUNIT)10000,18,19
   19 IF(PUNIT-1.1)20,10000,21
   20 PBAR=PBAR*0.4912
      GO TO 18
   21 PBAR=PBAR*0.03613
   18 IF(FLTYPE)10000,22,23
   22 CALL PRPS(TBAR+460.,PBAR+ATM,.25,RHO,HX,SX,XX)
      CALL TKCP(TBAR+460.,1.0/RHO,XK,CP)
      RAT=CP/(CP-0.98505)
      VIS=HMU(PBAR+ATM,1.0/RHO,TBAR+460.,0)
      GC=766.4
      GO TO 300
   23 RHO=2.6997*(PBAR+ATM)/(TBAR+460.)
      RAT=1.4
      VIS=(3.669E-07+5.56 E-10*(TBAR-50.))*32.2
      GC=53.33
  300 IF(PNUNIT)10000,24,25
   25 IF(PNUNIT-1.1)26,10000,27
   26 PUP=PUP*0.4912
      PDOWN=PDOWN*0.4912
      GO TO 24
   27 PUP=PUP*0.03613
      PDOWN=PDOWN*0.03613
   24 IF(ZONTYP)10000,28,29
   28 FLOW=C*(PUP+ATM)/SQRTF(TN+460.)
      GO TO 400
   29 R=(PDOWN+ATM)/(PUP+ATM)
      IF(FLTYPE)10000,50,51
   50 CALL PRPS(TN+460.,PUP+ATM,.25,RH1,HX,SX,XX)
      CALL TKCP(TN+460.,1.0/RH1,XK,CR)
      RA1=CR/(CR-0.98505)
      GO TO 10716
   51 RH1=2.6997*(PUP+ATM)/(TN+460.)
      RA1=1.4
10716 FLOW=(C/12.)*SQRTF(2.*RA1*RH1*32.2*(PUP+ATM)*(R**(2./RA1)-R**(1.+1
     1./RA1))/((RA1-1.)*(1.-BETASQ**2*R**(2./RA1))))
  400 IF(DPUNIT)10000,30,31
```

FORTRAN SOURCE PROGRAM

(Continued)

```
      DIMENSION COMM(40)                                                 8/31/64
   31 IF(DPUNIT-1.1)32,10000,33
   32 DPBAR=DPBAR*0.4912
      GO TO 30
   33 DPBAR=DPBAR*0.03613
   30 CONST=D*2.*RHO*32.2*DPBAR*AREA**2/(144.*FLOW**2)
      RE=48.*FLOW/(VIS*4.*AREA/EQDIA)
      U=144.*FLOW/(RHO*AREA*SQRTF(RAT*32.2*GC*(TBAR+460.)))
      WRITE OUTPUT TAPE 6,6,PBAR,TBAR,DPBAR,RHO,VIS,RAT,FLOW,CONST,RE,U
    6 FORMAT(5X,F5.1,4X,F5.1,4X,F5.1,4X,F6.3,4X,1PE11.4,4X,0PF6.3,4X,F5.
     13,4X,F6.3,4X,1PE11.4,4X,0PF5.3)
 1000 CONTINUE
      GO TO 5000
10000 STOP
      END
```

FFL-8B CONTROL DRUM ORIFICE   CALIBRATION OF CONTROL DRUM ANNULUS ON AIR

RUN NO. 1 .048 INCH ANNULUS JUNE 24,1964 INLET LOSS COEFFICIENT

AREA = 0.62852 EQUIVALENT DIAMETER = 0.09600

PAVE PSIG	TAVE DEG.F	DPAVE PSID	DENSITY LB/CUFT	VISCOSITY LB/FT-SEC	K	FLOW LB/SEC	CLOSS	REYNOLDS	MACH
85.2	80.5	1.1	0.497	1.2360E-05	1.400	0.110	7.911	1.6340E 04	0.045
84.7	81.0	2.2	0.495	1.2369E-05	1.400	0.154	8.052	2.2885E 04	0.063
85.2	82.5	3.3	0.496	1.2396E-05	1.400	0.189	7.973	2.7954E 04	0.077
85.5	80.5	4.2	0.499	1.2360E-05	1.400	0.217	7.785	3.2142E 04	0.087
84.7	79.5	5.1	0.496	1.2342E-05	1.400	0.240	7.760	3.5578E 04	0.097
84.7	79.0	5.9	0.496	1.2333E-05	1.400	0.263	7.512	3.9065E 04	0.107
84.7	78.0	6.7	0.497	1.2315E-05	1.400	0.281	7.456	4.1870E 04	0.114
85.2	77.0	7.7	0.501	1.2298E-05	1.400	0.302	7.512	4.4987E 04	0.122
85.5	76.5	8.3	0.502	1.2289E-05	1.400	0.318	7.301	4.7492E 04	0.128
84.7	77.0	8.9	0.498	1.2298E-05	1.400	0.330	7.179	4.9243E 04	0.134

Sample Output Sheet for Loss Coefficient Calculation

FFL-8B CONTROL DRUM ORIFICE TEST CALIBRATION OF DRUM ANNULUS ON AIR

RUN NO. 1 .048 INCH ANNULUS JUNE 24,1964 FRICTION FACTOR THRU ANNULUS

AREA = 1.02870 EQUIVALENT DIAMETER = 0.15600

PAVE PSIG	TAVE DEG.F	DPAVE PSID	DENSITY LB/CUFT	VISCOSITY LB/FT-SEC	K	FLOW LB/SEC	FF	REYNOLDS	MACH
84.2	80.5	0.2	0.493	1.2360E-05	1.400	0.110	0.014	1.6223E 04	0.027
82.5	81.0	0.4	0.483	1.2369E-05	1.400	0.154	0.012	2.2721E 04	0.039
82.0	82.5	0.7	0.479	1.2396E-05	1.400	0.189	0.013	2.7754E 04	0.048
81.2	80.5	0.8	0.478	1.2360E-05	1.400	0.217	0.012	3.1912E 04	0.056
80.2	79.5	1.0	0.473	1.2342E-05	1.400	0.240	0.012	3.5324E 04	0.062
79.0	79.0	1.4	0.468	1.2333E-05	1.400	0.263	0.013	3.8786E 04	0.069
78.2	78.0	1.7	0.465	1.2315E-05	1.400	0.281	0.015	4.1571E 04	0.075
77.7	77.0	2.2	0.463	1.2298E-05	1.400	0.302	0.016	4.4666E 04	0.080
77.0	76.5	2.0	0.460	1.2289E-05	1.400	0.318	0.013	4.7153E 04	0.085
76.0	77.0	2.2	0.454	1.2298E-05	1.400	0.330	0.013	4.8891E 04	0.090

Sample Output Sheet for Friction Factor Calculation